



PROYECTO FINAL

INSTITUTO TECNOLÓGICO AUTÓNOMO DE MÉXICO

PROGRAMACIÓN LINEAL

Proyecto 2

Antonio Bueno Royo

Adrián Janus González Adamúz

Profesora
Dra. Martha Cabo Nodar

1. Contexto del Problema

Nos han asignado la tarea de maximizar las ganancias de una fábrica que produce siete productos utilizando cinco tipos diferentes de máquinas. Cada producto aporta un beneficio específico a la empresa y requiere un tiempo determinado para su fabricación. Además, las máquinas necesitan mantenimiento anual, que generalmente se extiende por un mes. También hemos sido informados sobre restricciones en el mercado que limitan la producción de algunos productos a una cierta cantidad mensual.

Es importante tener en cuenta que la fábrica opera seis días a la semana, con dos turnos de trabajo de ocho horas cada uno. Con un mes típico de 24 días laborables, nos enfrentamos al desafío de optimizar la producción para maximizar los ingresos.

Para abordar este desafío, hemos desarrollado tres modelos distintos, cada uno más sofisticado que el anterior. Esta variedad de modelos nos permite adaptarnos a diferentes contextos de producción, brindando a la fábrica la flexibilidad necesaria para optimizar sus operaciones en diversas situaciones.

2. Modelo de AMPL

2.1. Modelo clásico

Nuestro primer modelo plantea un problema clásico de producción, que busca maximizar las ganancias de la empresa durante los primeros seis meses del año, sin considerar el almacenamiento de inventario. Es decir, partimos del supuesto de que la empresa logra vender todo su inventario. El enfoque matemático del problema se presenta a continuación:

Conjuntos

- *Prod*: Conjunto de productos i .
- *Mes*: Conjunto de meses j .
- *Maquina*: Conjunto de máquinas k .

Datos

- $horas_{k,i}$: Horas requeridas por la máquina k para producir el producto i .
- $beneficio_i$: Beneficio por unidad del producto i .
- $maquinas_k$: Cantidad de máquinas disponibles del tipo k .
- $ventas_{j,i}$: Máximo número de unidades del producto i que se pueden vender en el mes j .
- $manten_{j,k}$: Número de máquinas del tipo k en mantenimiento durante el mes j .

Variables

- $x_{i,j} \geq 0$: Cantidad de unidades del producto i producidas en el mes j .

Función Objetivo

$$\max_x \sum_{i \in Prod, j \in Mes} x_{i,j} \cdot beneficio_i \quad (1)$$

Restricciones

■ Horas disponibles por máquina:

$$\sum_{i \in Prod} horas_{j,i} \cdot x_{i,j} \leq 384 \times (maquinas_k - manten_{j,k}), \quad \forall j \in Mes, \forall k \in Maquina \quad (2)$$

Para comprender la condición (2), analicemos en términos de horas de trabajo de una máquina específica, por ejemplo, la máquina "m", dentro de un período de producción determinado "p". Esta restricción establece que el número de horas que esta máquina puede trabajar está limitado por la cantidad de unidades del producto "p" que se están produciendo en el mes "i". La cantidad de horas de trabajo no puede exceder el tiempo total de trabajo disponible durante el mes, definido como 2 jornadas por día, cada jornada de 8 horas, durante 24 días laborales, lo que da un total de 384 horas. Este límite se multiplica por un factor que considera la disponibilidad de las máquinas, es decir, el número total de máquinas menos aquellas que actualmente están en mantenimiento.

■ Limitación de mercado:

$$x_{i,j} \leq ventas_{j,i}, \quad \forall j \in Mes, \forall i \in Prod \quad (3)$$

La condición (3) se entiende que dado que tenemos restricciones de producción impuestas por el mercado es necesario tener que limitar la producción de unidades del producto p en el mes i .

Además, adjuntamos una captura de pantalla del modelo implementado en AMPL (Figura 3).

```

set Prod;
set Mes;
set Maquina;

param horas{Maquina, Prod};
param beneficio{Prod};
param maquinas{Maquina};
param ventas{Mes, Prod};
param manten{Mes, Maquina};

var x{Prod, Mes} >= 0;

maximize Beneficio_Mes:
    sum{i in Prod, j in Mes} x[i,j]*beneficio[i];

subject to Horas_prod_maquina{i in Mes, j in Maquina}:
    sum{k in Prod} horas[j,k]*x[k,i] <= 384*(maquinas[j]-manten[i,j]);
subject to Limitacion_mercado{i in Mes, j in Prod}:
    x[j,i] <= ventas[i,j];

```

Figura 1: Modelo clásico en AMPL

2.2. Extensión multiperiodo

El segundo modelo introduce la posibilidad de almacenar hasta 100 unidades de cada producto, con la particularidad de que el almacenamiento de cada unidad implica un costo mensual de \$0.5. Además, se establece como objetivo mantener un stock de 50 unidades de cada tipo al finalizar el mes de junio. Esta variante no solo amplía la comprensión del problema, sino que también proporciona la oportunidad de almacenar productos, lo que potencialmente conduce a mayores beneficios.

Conjuntos

- *Prod*: Conjunto de productos i .
- *Mes*: Conjunto de meses, ordenados, j .
- *Maquina*: Conjunto de máquinas, k .

Parámetros

- $horas_{k,i}$: Horas requeridas por la máquina k para producir el producto i .
- $beneficio_i$: Beneficio por unidad del producto i .
- $maquinas_k$: Cantidad de máquinas disponibles del tipo k .
- $ventas_{j,i}$: Máximo número de unidades del producto i que se pueden vender en el mes j .
- $manten_{j,k}$: Número de máquinas del tipo k en mantenimiento durante el mes j .

Variables

- $producido_{i,j} \geq 0$: Cantidad de unidades del producto i producidas en el mes j .
- $venta_{i,j} \geq 0$: Cantidad de unidades del producto i vendidas en el mes j .
- $guardado_{i,j} \geq 0$: Cantidad de unidades del producto i almacenadas al final del mes j .

Hemos desglosado nuestra variable original, "x", en tres nuevas variables: "guardado", "ventaz" "producido". Esta acción amplía el alcance de nuestra variable, ya que cada una de ellas representa una opción distinta en lugar de simplemente la cantidad vendida. Ahora, además de vender productos, tenemos la opción de guardar una parte de ellos, lo que brinda mayor flexibilidad en las operaciones comerciales.

Función Objetivo

$$\sum_{i \in Prod, j \in Mes} (venta[i, j] \cdot beneficio[i] - 0,5 \cdot guardado[i, j]) \quad (4)$$

Restricciones

■ **Horas disponibles por máquina:**

$$\sum_{i \in Prod} horas_{k,i} \cdot producido_{i,j} \leq 384 \times (maquinas_k - manten_{j,k}), \quad (5)$$

$$\forall k \in Mes, \forall k \in Maquina$$

La restricción (5) establece que la suma de las horas de trabajo utilizadas por cada máquina para producir productos no puede exceder un límite mensual predefinido, ajustado por el número de máquinas que están en mantenimiento. Se calcula multiplicando las horas necesarias para producir cada producto en una máquina específica por si se produce o no el producto durante el mes en cuestión, y esta suma debe ser menor o igual al total de horas disponibles por máquina en el mes multiplicado por las horas de la jornada laboral y los días hábiles ($8 \times 2 \times 24 = 384$), teniendo en cuenta cualquier máquina que esté en mantenimiento. Esto asegura una distribución equilibrada y eficiente de las horas de trabajo entre las máquinas, evitando exceder la capacidad de producción disponible.

■ **Limitación de mercado:**

$$venta_{i,j} \leq ventas_{j,k}, \quad \forall j \in Mes, \forall i \in Prod \quad (6)$$

La restricción (6) es idéntica a la condición (3)

■ **Condición de inventario:**

$$guardado_{i,j-1} + producido_{i,j} = venta_{i,j} + guardado_{i,j}, \quad (7)$$

$$\forall i \in Prod, \forall j \in Mes, \text{mes} \neq \text{primer mes, Enero}$$

La restricción (7) establece que la cantidad guardada al final del mes anterior más la cantidad producida en el mes actual debe ser igual a la suma de las ventas y la cantidad guardada al final del mes actual. Esta relación es crucial para garantizar un equilibrio entre la producción, las ventas y el inventario a lo largo del tiempo, evitando excesos o faltantes de productos. Además, se aplica a todos los productos y todos los meses, excepto el primer mes (Enero), lo que garantiza un control continuo del inventario en cada período de tiempo.

■ **Condición de almacenamiento nulo en Enero:**

$$producido_{i,Enero} = venta_{i,Enero} + guardado_{i,Enero}, \quad \forall i \in Prod \quad (8)$$

La restricción (8) se formula en respuesta a la suposición del problema de que no hay inventario almacenado antes de enero. Por lo tanto, la producción se iguala a las ventas menos el inventario guardado, lo que refleja la necesidad de mantener un equilibrio entre la producción, las ventas y el inventario en cada período.

■ **Límite de almacenamiento a 100 unidades:**

$$guardado_{i,j} \leq 100, \quad \forall j \in Mes, \forall i \in Prod \quad (9)$$

La restricción (9) se aplica con el propósito de restringir la cantidad de inventario disponible a la cantidad especificada en el problema

■ Condición de 50 unidades almacenadas en Junio:

$$guardado_{i,\text{Junio}} = 50, \quad \forall i \in \text{Prod} \quad (10)$$

Siguiendo la solicitud, se implementa la restricción (10) para garantizar que, para cada producto, el almacenamiento en junio sea de exactamente 50 unidades.

Además, adjuntamos una captura de pantalla del segundo modelo implementado en AMPL.

```
# Declaración de conjuntos
set Prod; # Define el conjunto de productos
set Mes ordered; # Define el conjunto de meses en orden
set Maquina; # Define el conjunto de máquinas

# Parámetros del modelo
param horas(Maquina, Prod); # Parámetro que indica las horas necesarias para producir cada producto en cada máquina
param beneficio(Prod); # Parámetro que indica el beneficio por producto
param maquinas(Maquina); # Parámetro que indica la cantidad de máquinas de cada tipo
param ventas(Mes, Prod); # Parámetro que indica las ventas esperadas de cada producto en cada mes
param manten(Mes, Maquina); # Parámetro que indica la cantidad de máquinas de cada tipo en mantenimiento en cada mes

# Variables de decisión
var producido(Prod, Mes) >= 0; # Cantidad de productos producidos en cada mes
var venta(Prod, Mes) >= 0; # Cantidad de productos vendidos en cada mes
var guardado(Prod, Mes) >= 0 <= 100; # Cantidad de productos guardados en cada mes, con límite de 100 unidades

# Función objetivo
maximize Beneficio_Mes:
    sum(i in Prod, j in Mes) (venta[i,j]*beneficio[i]-0.5*guardado[i,j]);
    # Maximiza el beneficio total considerando ventas y costos de almacenamiento

# Restricciones
subject to Horas_prod_maquina(i in Mes, j in Maquina):
    sum(k in Prod) horas[j,k]*producido[k,i] <= 384*(maquinas[j]-manten[i,j]);
    # Restringe la producción total de cada máquina en cada mes por su capacidad y el mantenimiento

subject to Limitacion_mercado(i in Mes, j in Prod):
    venta[j,i] <= ventas[i,j];
    # Limita las ventas de cada producto en cada mes según las expectativas de ventas

subject to junio_50(i in Prod):
    guardado[i, "Junio"] = 50;
    # Establece que al final de junio, el inventario de cada producto sea de 50 unidades

subject to condicion_inventario(i in Prod, j in Mes: j!=first(Mes)):
    guardado[i,prev(j)]+producido[i,j]=venta[i,j]+guardado[i,j];
    # Establece la relación entre el inventario guardado, la producción y las ventas para cada producto en cada mes, excepto en el primer mes

subject to Inicio(i in Prod):
    producido[i,"Enero"]=venta[i,"Enero"]+guardado[i,"Enero"];
    # Establece la producción inicial de cada producto en enero como la suma de las ventas y el inventario guardado en enero
```

Figura 2: Modelo multiperiodo en AMPL

2.3. Decisión de mantenimiento

El modelo multiperiodo busca maximizar el beneficio mensual considerando la producción, venta y almacenamiento de productos a lo largo del tiempo. La nueva variante introduce la flexibilidad de elegir el mejor mes para realizar el mantenimiento de cada máquina. En lugar de establecer un momento fijo para el mantenimiento de cada máquina, se busca determinar el mes óptimo para su inactividad por mantenimiento. Además, se especifica que cada máquina debe estar inactiva por mantenimiento en uno de los seis meses del periodo de planificación, excepto los molinos, donde solo dos de ellos deben recibir mantenimiento en el semestre.

Este enfoque amplía el modelo original al permitir una planificación más flexible de los mantenimientos de las máquinas, lo que potencialmente mejora la eficiencia del proceso al ajustar el tiempo de inactividad de las máquinas de acuerdo con las necesidades del sistema de producción.

Conjuntos

- *Prod*: Conjunto de productos *i*.

- *Mes*: Conjunto de meses, ordenados, j .
- *Maquina*: Conjunto de máquinas, k .

Parámetros

- $horas_{k,i}$: Horas requeridas por la máquina k para producir el producto i .
- $beneficio_i$: Beneficio por unidad del producto i .
- $maquinas_k$: Cantidad de máquinas disponibles del tipo k .
- $ventas_{j,i}$: Máximo número de unidades del producto i que se pueden vender en el mes j .
- $manten_{j,k}$: Número de máquinas del tipo k en mantenimiento durante el mes j .

Variables

- $producido_{i,j} \geq 0$: Cantidad de unidades del producto i producidas en el mes j .
- $venta_{i,j} \geq 0$: Cantidad de unidades del producto i vendidas en el mes j .
- $guardado_{i,j} \geq 0$: Cantidad de unidades del producto i almacenadas al final del mes j .
- $manten_{j,k} \geq 0$: Número de máquinas del tipo k en mantenimiento durante el mes j .

Las variables de producción, venta y almacenamiento permanecen idénticas al problema anterior. Se introduce la variable *manten*, la cual indica el número de máquinas del tipo m que están en mantenimiento durante el mes i , especificando el tiempo de inactividad planificado para cada tipo de máquina en un periodo específico. Estas variables son fundamentales para la planificación y optimización del proceso productivo, así como para la gestión eficiente del inventario y el mantenimiento de las máquinas.

Función Objetivo

$$\text{Maximizar Beneficio : } \sum_{i \in Prod, j \in Mes} (venta[i, j] \cdot beneficio[i] - 0,5 \cdot guardado[i, j]) \quad (11)$$

Restricciones

- **Horas disponibles por máquina:**

$$\sum_{i \in Prod} horas_{k,i} \cdot producido_{i,j} \leq 384 \times (maquinas_k - manten_{j,k}), \quad (12)$$

$$\forall k \in Mes, \forall k \in Maquina$$

- **Limitación de mercado:**

$$venta_{i,j} \leq ventas_{j,k}, \quad \forall j \in Mes, \forall i \in Prod \quad (13)$$

■ **Condición de inventario:**

$$guardado_{i,j-1} + producido_{i,j} = venta_{i,j} + guardado_{i,j}, \quad (14)$$

$$\forall i \in Prod, \forall j \in Mes, \text{mes} \neq \text{primer mes, Enero}$$

■ **Condición de almacenamiento nulo en Enero:**

$$producido_{i,Enero} = venta_{i,Enero} + guardado_{i,Enero}, \quad \forall i \in Prod \quad (15)$$

■ **Límite de almacenamiento a 100 unidades:**

$$guardado_{i,j} \leq 100, \quad \forall j \in Mes, \forall i \in Prod \quad (16)$$

■ **Condición de 50 unidades almacenadas en Junio:**

$$guardado_{i,Junio} = 50, \quad \forall i \in Prod \quad (17)$$

Las restricciones hasta este punto son idénticas al modelo anterior.

■ **Optimización del mantenimiento:**

$$\text{Sujeto a } mantenimiento_opti_k : \sum_{j \in Mes} manten[j,k] = opti[k], \quad \forall k \in Maquina \quad (18)$$

La restricción (18) nos permite imponer al problema un requisito mínimo de mantenimiento para cada máquina, asegurando que se realice al menos una vez por mes. Además, proporciona los meses óptimos para llevar a cabo estos mantenimientos de cada máquina, con el objetivo de maximizar la producción. Esta restricción es crucial para garantizar un funcionamiento adecuado de las máquinas a lo largo del tiempo, optimizando su rendimiento y minimizando posibles problemas operativos.

Además, adjuntamos una captura de pantalla del tercer modelo implementado en AMPL.

```
# Definición de conjuntos
set Prod;           # Conjunto de productos
set Mes ordered;    # Conjunto de meses, ordenado
set Maquina;        # Conjunto de máquinas

# Parámetros
param horas{Maquina, Prod};    # Horas requeridas para producir cada producto en cada máquina
param beneficio{Prod};         # Beneficio por producto
param maquinas{Maquina};       # Cantidad de máquinas disponibles
param ventas{Mes, Prod};       # Ventas esperadas de cada producto en cada mes
param opti{Maquina}>=0;         # Cantidad óptima de mantenimiento para cada máquina

# Variables de decisión
var producido{Prod, Mes}>=0;    # Producción de cada producto en cada mes
var venta{Prod, Mes}>=0;        # Ventas de cada producto en cada mes
var guardado{Prod, Mes}>=0 <=100; # Inventario de cada producto en cada mes
var manten{Mes, Maquina} integer >=0; # Cantidad de mantenimiento realizado en cada mes para cada máquina

# Función Objetivo
maximize Beneficio_Mes:
    sum{i in Prod, j in Mes} (venta[i,j]*beneficio[i]-0.5*guardado[i,j]);

# Restricciones
subject to Horas_prod_maquina{i in Mes, j in Maquina}:
    sum{k in Prod} horas[j,k]*producido[k,i] <= 384*(maquinas[j]-manten[i,j]);
    # Límite de horas de producción por máquina y mes

subject to Limitacion_mercado{i in Mes, j in Prod}:
    venta[j,i]<=ventas[i,j]; # Ventas no pueden exceder las ventas esperadas

subject to condicion_inventario{i in Prod, j in Mes: j!=first(Mes)}:
    guardado[i,prev(j)]+producido[i,j]=venta[i,j]+guardado[i,j]; # Actualización del inventario

subject to Inicio{i in Prod}:
    producido[i,"Enero"]=venta[i,"Enero"]+guardado[i,"Enero"]; # Inventario inicial en enero

subject to junio_50{i in Prod}:
    guardado[i, "Junio"]=50; # Fijar el inventario en junio

subject to mantenimiento_opti{m in Maquina}:
    sum{j in Mes} manten[j,m]=opti[m]; # Mantenimiento óptimo por máquina
```

Figura 3: Modelo decisión de mantenimiento en AMPL

3. Soluciones

Todos los resultados presentados a continuación se basaron en los datos proporcionados en la sección Datos del Modelo, ubicada al final del documento. Estos mismos datos fueron utilizados tanto en la formulación del modelo clásico como en la extensión multiperíodo, con ligeras modificaciones para el modelo de Decisión de Mantenimiento.

3.1. Solución del modelo clásico

	Enero	Febrero	Marzo	Abril	Mayo	Junio	Beneficios por producto
PROD 1	500	600	0	200	0	500	18000
PROD 2	888.571	500	0	300	100	500	13731.426
PROD 3	300	200	0	400	500	0	11200
PROD 4	300	0	0	500	100	300	4800
PROD 5	800	400	0	200	1000	0	26400
PROD 6	200	300	400	0	300	500	15300
PROD 7	0	150	0	100	0	0	750
Beneficios mensuales	24531.426	18150	3600	11500	18700	13700	90181.426

Cuadro 1: Datos mensuales para diferentes categorías con costos totales.

El valor óptimo de beneficios de la fábrica asciende a \$90,181.426. Para interpretar la tabla, tomemos el ejemplo de la producción de febrero del producto 7, que alcanzó las 150 unidades con un beneficio mensual de \$18,150. Esta lectura cobra sentido dentro del contexto clásico de planteamiento, dado que la totalidad del producto fue vendida.

3.2. Soluciones del modelo de extensión multiperíodo

Cuadro 2: Producción, venta y guardado por mes con Costes totales.

Mes	Producción	Venta	Guardado
Enero	500 (PROD 1)	500 (PROD 1)	0 (PROD 1)
	888.571 (PROD 2)	888.571 (PROD 2)	0 (PROD 2)
	382.50 (PROD 3)	300 (PROD 3)	82.50 (PROD 3)
	300 (PROD 4)	300 (PROD 4)	0 (PROD 4)
	800 (PROD 5)	800 (PROD 5)	0 (PROD 5)
	200 (PROD 6)	200 (PROD 6)	0 (PROD 6)
	0 (PROD 7)	0 (PROD 7)	0 (PROD 7)
Costes enero	25191.426	24531.426	660
Beneficios enero	24531.426		
Febrero	700 (PROD 1)	600 (PROD 1)	100 (PROD 1)
	600 (PROD 2)	500 (PROD 2)	100 (PROD 2)
	117.50 (PROD 3)	200 (PROD 3)	0 (PROD 3)
	0 (PROD 4)	0 (PROD 4)	0 (PROD 4)

Mes	Producción	Venta	Guardado
	500 (PROD 5)	400 (PROD 5)	100 (PROD 5)
	300 (PROD 6)	300 (PROD 6)	0 (PROD 6)
	250 (PROD 7)	150 (PROD 7)	100 (PROD 7)
Costes febrero	20490	18150	3000
Beneficios febrero	18150		
Marzo	0 (PROD 1)	100 (PROD 1)	0 (PROD 1)
	0 (PROD 2)	100 (PROD 2)	0 (PROD 2)
	0 (PROD 3)	0 (PROD 3)	0 (PROD 3)
	0 (PROD 4)	0 (PROD 4)	0 (PROD 4)
	0 (PROD 5)	100 (PROD 5)	0 (PROD 5)
	400 (PROD 6)	400 (PROD 6)	0 (PROD 6)
	0 (PROD 7)	100 (PROD 7)	0 (PROD 7)
Costes marzo	3600	6600	0
Beneficios marzo	6600		
Abril	200 (PROD 1)	200 (PROD 1)	0 (PROD 1)
	300 (PROD 2)	300 (PROD 2)	0 (PROD 2)
	400 (PROD 3)	400 (PROD 3)	0 (PROD 3)
	500 (PROD 4)	500 (PROD 4)	0 (PROD 4)
	200 (PROD 5)	200 (PROD 5)	0 (PROD 5)
	0 (PROD 6)	0 (PROD 6)	0 (PROD 6)
	100 (PROD 7)	100 (PROD 7)	0 (PROD 7)
Costes abril	11500	11500	0
Beneficios abril	11500		
Mayo	0 (PROD 1)	0 (PROD 1)	0 (PROD 1)
	100 (PROD 2)	100 (PROD 2)	0 (PROD 2)
	600 (PROD 3)	500 (PROD 3)	100 (PROD 3)
	100 (PROD 4)	100 (PROD 4)	0 (PROD 4)
	1100 (PROD 5)	1000 (PROD 5)	100 (PROD 5)
	300 (PROD 6)	300 (PROD 6)	0 (PROD 6)
	100 (PROD 7)	0 (PROD 7)	100 (PROD 7)
Costes mayo	20900	18700	2200
Beneficios mayo	18700		
Junio	550 (PROD 1)	500 (PROD 1)	50 (PROD 1)
	550 (PROD 2)	500 (PROD 2)	50 (PROD 2)
	0 (PROD 3)	50 (PROD 3)	50 (PROD 3)
	350 (PROD 4)	300 (PROD 4)	50 (PROD 4)
	0 (PROD 5)	50 (PROD 5)	50 (PROD 5)
	550 (PROD 6)	500 (PROD 6)	50 (PROD 6)
	0 (PROD 7)	50 (PROD 7)	50 (PROD 7)
Costes junio	15150	14800	2550
Beneficios junio	13525		
Costes totales	96831.426	94281.426	8410
Beneficios totales	93006.426		

El valor óptimo de beneficios de la fábrica alcanza los \$93,715.18. Además, es crucial entender la tabla de la siguiente manera: en mayo, la producción del producto 5 fue de 1,100 unidades, de las cuales se vendieron 1,000 unidades y se reservaron 50. Observa-

mos también que el modelo sugiere incrementar el almacenamiento a medida que nos aproximamos a junio, mientras que en marzo y abril no se reservó nada, posiblemente debido al mantenimiento de las máquinas. En marzo, se eliminó la restricción que limitaba la producción a solo dos productos. Por último, es importante destacar que este modelo generó un aumento del 3.9% en ganancias en comparación con el anterior

3.3. Soluciones del modelo de Decisión de mantenimiento

Cuadro 3: Máquinas en mantenimiento por mes.

Mes	Garlopa	Molinos	Perforadora	Taladros_H	Taladros_V
Enero	-	-	-	1	-
Febrero	-	-	-	-	-
Marzo	-	-	-	-	-
Abril	1	-	1	-	-
Mayo	-	2	-	-	1
Junio	-	-	-	-	-

La tabla indica el número de máquinas que están en mantenimiento en cada mes. Si en una fila aparece un guión, esto significa que ninguna máquina de ese tipo se encuentra en mantenimiento durante ese mes.

Cuadro 4: Producción, venta, guardado por mes con costes totales.

Mes	Producción	Venta	Guardado
Enero	500 (PROD 1)	500 (PROD 1)	0 (PROD 1)
	1000 (PROD 2)	1000 (PROD 2)	0 (PROD 2)
	300 (PROD 3)	300 (PROD 3)	0 (PROD 3)
	300 (PROD 4)	300 (PROD 4)	0 (PROD 4)
	800 (PROD 5)	800 (PROD 5)	0 (PROD 5)
	200 (PROD 6)	200 (PROD 6)	0 (PROD 6)
	100 (PROD 7)	100 (PROD 7)	0 (PROD 7)
Costes Enero	25500	25500	0
Beneficios enero	25500		
Febrero	600 (PROD 1)	600 (PROD 1)	0 (PROD 1)
	500 (PROD 2)	500 (PROD 2)	0 (PROD 2)
	200 (PROD 3)	200 (PROD 3)	0 (PROD 3)
	0 (PROD 4)	0 (PROD 4)	0 (PROD 4)
	400 (PROD 5)	400 (PROD 5)	0 (PROD 5)
	300 (PROD 6)	300 (PROD 6)	0 (PROD 6)
	150 (PROD 7)	150 (PROD 7)	0 (PROD 7)
Costes Febrero	18150	18150	0
Beneficios febrero	18150		
Marzo	400 (PROD 1)	300 (PROD 1)	100 (PROD 1)
	700 (PROD 2)	600 (PROD 2)	100 (PROD 2)
	100 (PROD 3)	0 (PROD 3)	100 (PROD 3)
	100 (PROD 4)	0 (PROD 4)	100 (PROD 4)
	600 (PROD 5)	500 (PROD 5)	100 (PROD 5)

Mes	Producción	Venta	Guardado
	400 (PROD 6)	400 (PROD 6)	0 (PROD 6)
	200 (PROD 7)	100 (PROD 7)	100 (PROD 7)
Costes Marzo	20200	16000	4200
Beneficios marzo	16000		
Abril	0 (PROD 1)	100 (PROD 1)	0 (PROD 1)
	0 (PROD 2)	100 (PROD 2)	0 (PROD 2)
	0 (PROD 3)	100 (PROD 3)	0 (PROD 3)
	0 (PROD 4)	100 (PROD 4)	0 (PROD 4)
	0 (PROD 5)	100 (PROD 5)	0 (PROD 5)
	0 (PROD 6)	0 (PROD 6)	0 (PROD 6)
	0 (PROD 7)	100 (PROD 7)	0 (PROD 7)
Costes Abril	0	4200	0
Beneficios abril	4200		
Mayo	0 (PROD 1)	0 (PROD 1)	0 (PROD 1)
	100 (PROD 2)	100 (PROD 2)	0 (PROD 2)
	500 (PROD 3)	500 (PROD 3)	0 (PROD 3)
	100 (PROD 4)	100 (PROD 4)	0 (PROD 4)
	1000 (PROD 5)	1000 (PROD 5)	0 (PROD 5)
	300 (PROD 6)	300 (PROD 6)	0 (PROD 6)
	0 (PROD 7)	0 (PROD 7)	0 (PROD 7)
Costes Mayo	18700	18700	0
Beneficios mayo	18700		
Junio	550 (PROD 1)	500 (PROD 1)	50 (PROD 1)
	550 (PROD 2)	500 (PROD 2)	50 (PROD 2)
	150 (PROD 3)	100 (PROD 3)	50 (PROD 3)
	350 (PROD 4)	300 (PROD 4)	50 (PROD 4)
	1150 (PROD 5)	1100 (PROD 5)	50 (PROD 5)
	550 (PROD 6)	500 (PROD 6)	50 (PROD 6)
	110 (PROD 7)	60 (PROD 7)	50 (PROD 7)
Costes Junio	29330	26780	2550
Beneficios junio	25505		
Costes Totales	111880	109330	6750
Beneficios totales	108055		

El valor óptimo de beneficios de la fábrica se sitúa en \$108,855. En cuanto a la interpretación de los resultados, sigue siendo similar, pero se ha añadido una tabla de mantenimiento que indica en qué mes se determinó que sería óptimo llevar a cabo el mantenimiento de cada máquina. Destaca que este modelo ha incrementado las ganancias en un 20.7%. De manera inesperada, se estipuló que en abril no se debía realizar ninguna producción, sino simplemente vender el inventario existente.

4. Interpretación

En esta sección, presentamos brevemente algunas conclusiones e interpretaciones tanto de los modelos como del problema en general. Es importante destacar que los tres problemas abordan la misma situación: una fábrica que busca maximizar sus beneficios. A simple vista, podría parecer que el último modelo es el más adecuado en cualquier situación, ¿verdad? Sin embargo, cada modelo tiene su lugar y su aplicación específica, dependiendo del contexto en el que se quiera implementar.

Por ejemplo, si nos encontramos en una situación donde vendemos suministros al gobierno y este siempre compra todo lo que producimos, entonces el primer modelo podría representar mejor esta situación. Por otro lado, si nos enfrentamos a un escenario donde se permite almacenar productos, pero los mantenimientos son fijos debido a cuestiones técnicas de las máquinas o a la disponibilidad de personal para el mantenimiento en determinados meses, entonces sería más conveniente utilizar el segundo modelo, aunque ambos casos puedan generar menos beneficios que el último.

Como nota general, consideramos que el último modelo es el más versátil y completo, ya que ofrece la mayor flexibilidad para modificar el modelo según las necesidades específicas. Esto se debe a que el último modelo, al introducir una variable adicional, en nuestro caso *manten*, agrega una dimensión adicional al problema, lo que aumenta significativamente el número de posibles soluciones. Esta mayor complejidad facilita la búsqueda de una solución óptima más ajustada a las condiciones y restricciones del problema.

Es importante tener en cuenta que este proceso de aumentar la flexibilidad del modelo al convertir parámetros en variables puede repetirse con otros aspectos del problema, pero es crucial considerar el aumento en el costo computacional que esto conlleva. A medida que se añaden más variables, el tiempo y los recursos necesarios para resolver el problema aumentan drásticamente. Por lo tanto, la elección del modelo adecuado dependerá de un equilibrio entre la flexibilidad deseada y la capacidad computacional disponible.

Como nota adicional, es importante tener en cuenta que AMPL puede presentar errores de redondeo en algunos casos, lo que puede ocasionar que los números no sumen exactamente de la misma manera que si se realizaran los cálculos manualmente. Sin embargo, es importante destacar que el margen de error suele ser pequeño.

Por lo tanto, si es necesario, se recomienda revisar los números con detalle y compararlos con los cálculos manuales. Si no coinciden, es posible que sea debido a los errores de redondeo inherentes al software AMPL.

5. Datos del Modelo

En esta sección presentamos imágenes de los datos utilizados en cada uno de los modelos implementados en AMPL.

Los datos que fueron utilizados en los primeros dos modelos presentados en este trabajo son idénticos. A continuación presentamos una imagen de estos.

```

set Prod := x1 x2 x3 x4 x5 x6 x7;
set Mes:= Enero Febrero Marzo Abril Mayo Junio;
set Maquina:= Molino T_vert T_hor Perf Garlopa;

param horas:
    x1    x2    x3    x4    x5    x6    x7 :=
Molino  0.5    0.7    0.0    0.0    0.3    0.2    0.5
T_vert  0.1    0.2    0.0    0.3    0.0    0.6    0.0
T_hor   0.2    0.0    0.8    0.0    0.0    0.0    0.6
Perf    0.05   0.03   0.0    0.07   0.1    0.0    0.08
Garlopa 0.0    0.0    0.01   0.0    0.05   0.0    0.05;

param beneficio:
    x1    x2    x3    x4    x5    x6    x7 :=
benef  10    6    8    4    11    9    3;

param manten:
    Molino T_vert T_hor Perf Garlopa :=
Enero    1      0      0      0      0
Febrero  0      0      2      0      0
Marzo    0      0      0      1      0
Abril    0      1      0      0      0
Mayo     1      1      0      0      0
Junio    0      0      1      0      1;

param maquinas:
    Molino T_vert T_hor Perf Garlopa :=
maq       4      2      3      1      1;

param ventas:
    x1    x2    x3    x4    x5    x6    x7 :=
Enero    500   1000   300   300   800   200   100
Febrero   600   500   200    0   400   300   150
Marzo     300   600    0    0   500   400   100
Abril     200   300   400   500   200    0   100
Mayo       0   100   500   100  1000   300    0
Junio     500   500   100   300  1100   500   60;

```

Figura 4: Datos de los primeros dos modelos en AMPL.

Los datos del tercer modelo se muestran en la siguiente imagen.

```

# Conjuntos
set Prod := x1 x2 x3 x4 x5 x6 x7;
set Mes := Enero Febrero Marzo Abril Mayo Junio;
set Maquina := Molino T_vert T_hor Perf Garlopa;

# Parámetros
param horas:
    x1    x2    x3    x4    x5    x6    x7 :=
    Molino 0.5    0.7    0.0    0.0    0.3    0.2    0.5
    T_vert 0.1    0.2    0.0    0.3    0.0    0.6    0.0
    T_hor  0.2    0.0    0.8    0.0    0.0    0.0    0.6
    Perf   0.05   0.03   0.0    0.07   0.1    0.0    0.08
    Garlopa 0.0    0.0    0.01   0.0    0.05   0.0    0.05;

param beneficio:
    x1    x2    x3    x4    x5    x6    x7 :=
    benef 10    6    8    4    11    9    3;

param opti:
    Molino T_vert T_hor Perf Garlopa :=
    minimo 2    1    1    1    1;

param maquinas:
    Molino T_vert T_hor Perf Garlopa :=
    maq    4    2    3    1    1;

param ventas:
    x1    x2    x3    x4    x5    x6    x7 :=
    Enero 500 1000 300 300 800 200 100
    Febrero 600 500 200 0 400 300 150
    Marzo 300 600 0 0 500 400 100
    Abril 200 300 400 500 200 0 100
    Mayo 0 100 500 100 1000 300 0
    Junio 500 500 100 300 1100 500 60;

```

Figura 5: Datos del tercer modelo en AMPL.